ODE mean field model for SIS network

```
> restart:with(LinearAlgebra):with(VectorCalculus):
```

RHS of ODE, using J fo rnumber of infected to distignuish from imaginary unit

```
> S:=1-J;
  lSI:=k/2-lSS-lII;
  sys:=Vector([-r*J+p*lSI,p*lSI*(lSI/S+1)-2*r*lII,(r+w)*lSI-2*p*
  lSI*lSS/S]);
```

$$S := 1 - J$$

$$lSI := \frac{1}{2}\,k - lSS - lII$$

$$sys := \left(-r\,J + p\left(\frac{1}{2}\,k - lSS - lII\right)\right)e_x + \left(p\left(\frac{1}{2}\,k - lSS - lII\right)\left(\frac{\frac{1}{2}\,k - lSS - lII}{1-J} + 1\right) - 2\,r\,lII\right)e_y + \left((r+w)\left(\frac{1}{2}\,k - lSS - lII\right) - \frac{2\,p\left(\frac{1}{2}\,k - lSS - lII\right)lSS}{1-J}\right)e_z \tag{1}$$

endeq are [LII,LSS,p] component of endemic equilibrium, parametrized by J

```
> endeqsol:=solve({sys[1],sys[2],sys[3]},{p,lSS,lII}):
  endeq:=subs(endeqsol,[lII,lSS,p]);factor(endeq[3]);
```

$$endeq := \left[-\frac{1}{2}\,\frac{J\left(-k\,r\,J + J^2\,r - r - w + 2\,w\,J - J^2\,w\right)}{r + J^2\,w + w - 2\,w\,J},\right.$$
$$\left.-\frac{1}{2}\,\frac{(-r + r\,J - w + w\,J)\left(-k\,J + k - J + J^2\right)}{r + J^2\,w + w - 2\,w\,J}, \frac{r + J^2\,w + w - 2\,w\,J}{-k\,J + k - J + J^2}\right]$$

$$\frac{r + J^2\,w + w - 2\,w\,J}{(J-k)\,(-1+J)} \tag{2}$$

Determine location of transcritical bifurcation by checking endemic equilibrium at J=0

```
> endeq0:=simplify(eval(endeq,J=0));
```

$$endeq0 := \left[0, \frac{1}{2}\,k, \frac{r+w}{k}\right] \tag{3}$$

> Check derivative of endemic equilibrium to check criticality of transcritical bifurcation (is endemic equilibrium stable or unstable?

```
  simplify(eval(diff(endeq[3],J),J=0));
```

$$\frac{-k\,w + k\,r + r + w}{k^2} \tag{4}$$

> For the concrete values of k=10, w=0.04 and r=0.002 this is (giving the plot below)

```
  pI:=eval(endeq[3],{r=0.002,w=0.04,k=10});
  plot([pI,J,J=0..0.98],view=[0..0.008,0..1],labels=['p','I']);
```

$$pI := \frac{0.042 + 0.04\,J^2 - 0.08\,J}{-11\,J + 10 + J^2}$$



> The fold can be obtained by simply checking where the derivative of p wrt J is zero. We parametrize the fold curve in the (p,w) plane by J. The plot below is the fold curve in (p,w,J)-space.

```
wfold:=solve(diff(endeq[3],J)=0,w);
pfold:=simplify(eval(endeq[3],w=wfold));
fc:=eval([pfold,wfold],{k=10,r=0.002});
plots[spacecurve]([fc[1],fc[2],J,J=0..0.9],view=[0..0.003,0..0.05,0..1],axes=normal,labels=['p','w','I']);
```

$$wfold := -\frac{r\,(-k-1+2\,J)}{(-1+J)^2\,(-1+k)}$$

$$pfold := -\frac{2\,r}{(-1+J)\,(-1+k)}$$

$$fc := \left[-\frac{0.0004444444444}{-1+J},\ -\frac{0.0002222222222\,(-11+2\,J)}{(-1+J)^2}\right]$$



> The Jacobian in the endemic equilbirium is
endJac:=simplify(eval(Jacobian(sys,[J,III,ISS]),endeqsol));

$$endJac := \left[\left[-r, -\frac{r+J^2 w+w-2 w J}{-k J+k-J+J^2}, -\frac{r+J^2 w+w-2 w J}{-k J+k-J+J^2}\right], \left[\frac{(J-k) r^2 J^2}{(r+J^2 w+w-2 w J)(-1+J)}, -\frac{J^2 w-2 r J-2 w J+r+2 k r+w}{(-1+J)(J-k)}, \frac{2 J^2 r-J^2 w+2 w J-2 k r J-r-w}{(-1+J)(J-k)}\right], \left[\frac{(r+w)(J-k) r J}{r+J^2 w+w-2 w J}, 0, \frac{2 r J}{-1+J}\right]\right] \quad (5)$$

> Its characteristic polynomial is
endJacp:=CharacteristicPolynomial(endJac,lambda);

$$endJacp := \lambda^3 - \frac{(J^2 r-k r J-J^2 w+3 r J+2 w J-r-3 k r-w)\lambda^2}{(-1+J)(J-k)} \quad (6)$$

$$-\frac{r\left(-rJ-wJ-3J^2r+2kr+3krJ-kwJ+r+w+kJ^2w\right)\lambda}{(-1+J)^2(J-k)}$$

$$-\frac{2Jr^2\left(kJ^2w-J^2w+2wJ+2rJ-2kwJ+kw-kr-r-w\right)}{(-1+J)^3(J-k)}$$

Gross et al (2006) observed a Hopf bifurcation above a certain minimal w. We only calculate this minimal w by checking for a double-eigenvalue zero of the Jacobian. At wdpzero dp/dlambda(0)=0. This has to be satisfied simultaneously with p(0)=0. The resulting equation has 3 solutions for J, only one of which is between 0 and 1.

```
> wdpzero:=solve(eval(diff(endJacp,lambda),lambda=0),w);
  solve(wfold=wdpzero,J);
  J00:=[%][1];
  wHopfmin:=evalf(eval(eval(wfold,J=J00),{k=10,r=0.002}));
```

$$wdpzero := \frac{r\left(J+3J^2-2k-3kJ-1\right)}{(-1+J)(kJ-1)}$$

$$\frac{1}{3}\frac{-1+\sqrt{1-6k+6k^2}}{-1+k},\ -\frac{1}{3}\frac{1+\sqrt{1-6k+6k^2}}{-1+k},\ k$$

$$J00 := \frac{1}{3}\frac{-1+\sqrt{1-6k+6k^2}}{-1+k}$$

$$wHopfmin := 0.06740853313 \tag{7}$$